Master 2 Humanités numériques – Bases de données semi-structurées

Examen sur machine – 19/01/2026

J. Darmont – https://eric.univ-lyon2.fr/jdarmont/

Durée : 2 heures – Documents autorisés sauf IAG – Barème fourni à titre indicatif

Rendu : Fichiers DTD (.dtd) et XQuery (.xq) sur Moodle.
https://moodle.univ-lyon3.fr/course/view.php?id=20834

Soient les deux documents XML *projects.xml* (projets) et *employees.xml* (employé·es) annexés dans Moodle.

Exercice 1 : DTD (10 points)

Définir pour chacun des deux documents XML une DTD permettant de valider les documents avec https://www.xmlvalidation.com.

Exercice 2 : XQuery (10 points)

Formuler les requêtes suivantes, en privilégiant la lisibilité du code XQuery. Tout résultat de requête doit être un fragment de code XML bien formé.

1. Nombre d'employé·es.

2. Salaire (*salary*) minimum et maximum des employé·es.

3. Nom (*firstName*) et prénom *(lastName)* des employé·es par ordre alphabétique de nom et de prénom.

4. Toutes les données du projet n° P03.

5. Nom des postes (*position*) dont le salaire est supérieur ou égal à 80000.

RSVP

6. Année de fondation de l'entreprise.

7. Noms, prénoms des employé·es et nom (*name*) du délivrable (*deliverable*) des projets correspondants.

8. Salaire moyen par département (*department*), trié par ordre inverse du salaire moyen.

9. Salaire maximum par département et par année, trié par ordre inverse d'année.

10. Nom des employé·es selon qu'ils sont managers ou employé·es. La liste des managers est :
    - Software Engineer
    - Project Manager
    - Data Analyst
    - HR Specialist
    - Business Analyst
    - HR Manager
    - Legal Advisor.

Correction Exercice 1

```
<!-- projects -->
<!ELEMENT projects (project*)>
  <!ELEMENT project (deliverable)>
  <!ATTLIST project id ID #REQUIRED>
    <!ELEMENT deliverable (name, output, publication)>
      <!ELEMENT name (#PCDATA)>
      <!ELEMENT output (#PCDATA)>
      <!ELEMENT publication (param*)>
        <!ELEMENT param (#PCDATA)>

<!-- employees -->
<!ELEMENT employees (employee*)>
  <!ELEMENT employee (id, firstName, lastName, position, department, hireDate,
                      salary, project)>
    <!ELEMENT id (#PCDATA)>
    <!ELEMENT firstName (#PCDATA)>
    <!ELEMENT lastName (#PCDATA)>
    <!ELEMENT position (#PCDATA)>
    <!ELEMENT department (#PCDATA)>
    <!ELEMENT hireDate (#PCDATA)>
    <!ELEMENT salary (#PCDATA)>
    <!ELEMENT proj (#PCDATA)>
```

Correction Exercice 2

```
(: 1 :)
let $c := count(//employee)
return <res>{$c}</res>

(: 2 :)
let $e := //employee/salary,
      $m := min($e),
      $M := max($e)
return <res><min>{$m}</min> <max>{$M}</max></res>

(: 3 :)
for $e in //employee
order by $e/lastName, $e/firstName
return <res>{$e/firstName} {$e/lastName}</res>

(: 4 :)
for $p in //project
where $p/@id = "P03"
return $p/*

(: 5 :)
for $e in //employee
where $e/salary >= 80000
return <res>{$e/position} {$e/salary}</res>
```

```
(: 6 :)
let $m := min(
    for $e in //employee
    return (year-from-date($e/hireDate))
)
return <foundation>{$m}</foundation>

(: 7 :)
for $e in //employee,
    $p in //project
where $e/proj = $p/@id
return <res>{$e/lastName} {$e/firstName} {$p//name}</res>

(: 8 :)
for $e in //employee
group by $g := $e/department
let $salmoy := avg($e/salary)
order by number($salmoy) descending
return <group department="{$g}">
           <salaire-moyen>{$salmoy}</salaire-moyen>
       </group>

(: 9 :)
for $e in //employee
group by $g := $e/department,
         $a := $e/year-from-date(hireDate)
let $M := max($e/salary)
order by $a descending
return <group department="{$g}" year="{$a}">
           <max-sal>{$M}</max-sal>
       </group>

(: 10 :)
for $p in //employee
return if ($p/position = "Software Engineer" or
           $p/position = "Project Manager" or
           $p/position = "Data Analyst" or
           $p/position = "HR Specialist" or
           $p/position = "Business Analyst" or
           $p/position = "HR Manager" or
           $p/position = "Legal Advisor")
       then <manager>{$p/lastName}</manager>
       else <employee>{$p/lastName}</employee>
```